# Difference Approximations of One Way Elastic Waves

Bjorn Engquist and Robert Clayton

Extrapolation of elastic waves in space is described by partial differential equations that can be written in the following form (SEP-10, p. 126).

$$U_{tz} + A\,U_{xz} + B\,U_{tt} + C\,U_{tx} + D\,U_{xx} = 0 \qquad (1)$$

$U(t, x, 0)$, $U(0, x, z)$ and $U_t(0, x, z)$ are given.

Here, A, B, C and D are 2x2 matrices. For the purposes of this paper the vector valued function U has the components $u(t,x,z)$ and $w(t,x,z)$ (the horizontal and vertical displacements respectively), but other dependent variables also give rise to equations of the form (1).

Equation (1) is in this approach solved with the space variable $z$ as the evolution direction and initial values are given for $z=0$. It is the vector analogue of the Claerbout scalar wave extrapolation, and hence differs conceptually from the discretization of an elastic wave equation with time as the evolution direction (see Kelley et al for example). In this paper we will derive difference approximations for (1) and also describe some results of numerical calculations.

The vector valued function $U(t,x,z)$ is approximated by the mesh function $U^n_{j,k}$ on a grid $(t_j, x_k, z^n)$, $j=0,1,\ldots J$, $k=0,1,\ldots K$, $n=0,1,\ldots$; $t_j = j\Delta t$ ; $x_k = k\Delta x$ and $z^n = n\Delta z$ . The main principles we will follow in the design of the difference scheme for (1) are:

1) to avoid $x$ differencing in the latest $t$ and $z$ levels in

order to get an explicit scheme.

2) to use centered differencing as much as possible for accuracy.

3) to avoid differencing that would cause instabilities in simple scalar model cases.

The difference operators $D_+$ , $D_-$ and $D_0$ were introduced in earlier reports (see for example, SEP-7, p. 136). They denote forward, backward and centered divided differences. For example,

$$D_+^z U_{j,k}^n = ( U_{j,k}^{n+1} - U_{j,k}^n ) / \Delta z \qquad ( \sim \frac{\partial}{\partial z} )$$

$$D_0^t U_{j,k}^n = ( U_{j+1,k}^n - U_{j-1,k}^n ) / 2 \Delta t \qquad ( \sim \frac{\partial}{\partial t} )$$

$$D_+^x D_-^x U_{j,k}^n = D_+^x ( D_-^x U_{j,k}^n ) =$$

$$= ( U_{j,k+1}^n - 2 U_{j,k}^n + U_{j,k-1}^n ) / \Delta x^2 \qquad ( \sim \frac{\partial^2}{\partial x^2} )$$

The derivatives in (1) can now be replaced by these difference operators or averages of them. We can think of the scheme as being centered around $( t_j , x_k , z^{n+1/2} )$ .

$$D_0^t D_+^z U_{j,k}^n + A D_0^x D_+^z U_{j,k}^n + \frac{1}{2} B D_+^t D_-^t ( U_{j,k}^n + U_{j,k}^{n+1} ) +$$

$$+ \frac{1}{2} C D_+^t D_0^x ( U_{j,k}^n + U_{j-1,k}^{n+1} ) + \frac{1}{2} D D_+^x D_-^x ( U_{j+1,k}^n + U_{j-1,k}^{n+1} ) = 0 \qquad (2)$$

$U_{j,k}^0$ , $U_{0,k}^n$ and $U_{1,k}^n$ are given.

Let us display the difference molecules in the $(t,z)$ plane




The scheme given in (2) will then look like

$$\left( \frac{I}{2\ \Delta t \Delta z} \begin{array}{|c|c|}\hline 1 & -1 \\ \hline & \\ \hline -1 & 1 \\ \hline \end{array} + \frac{A}{2\ \Delta x \Delta z} \begin{array}{|c|c|}\hline & \\ \hline -1 & 1 \\ \hline & \\ \hline \end{array} (2\ \Delta x\, D_0^x) \right. +$$

$$+ \frac{B}{2\ \Delta t^2} \begin{array}{|c|c|}\hline 1 & 1 \\ \hline -2 & -2 \\ \hline 1 & 1 \\ \hline \end{array} + \frac{C}{4\ \Delta t \Delta x} \begin{array}{|c|c|}\hline & -1 \\ \hline -1 & 1 \\ \hline 1 & \\ \hline \end{array} (2\ \Delta x\, D_0^x)$$

$$\left. + \frac{D}{2\ \Delta x^2} \begin{array}{|c|c|}\hline & 1 \\ \hline & \\ \hline 1 & \\ \hline \end{array} (\Delta x^2\ D_+^x\, D_-^x) \right) U = 0 \qquad (3)$$

Let us finally present the scheme in a third way which is closer to the formulation used when programming.

The differencing scheme of (3) can be cast in the form of eight (3x3) pattern matrices $E^+$, $E^-$, $F^+$, $F^-$, $G^+$, $G^-$, $H^+$ and $H^-$, which describe the net effect of all the differentials at the point ( j, k, $n+\frac{1}{2}$ ) in both the u and w fields. The E and F matrices correspond to the coefficients of the first component of equation (2) . They denote the dependence of that component on u and w respectively. Similarly, G and H denote the dependence of the second component of equation (2) on w and u respectively. The superscripts indicate whether the matrix is applied to the current z level ('+') or the previous z level ('-') .

The pattern for the 'E' matrices is

$E^-$

| | k-1 | k | k+1 |
|---|---|---|---|
| j-1 | 0 | $e_1$ | 0 |
| j | $e_2$ | $e_3$ | $-e_2$ |
| j+1 | $e_4$ | $e_5$ | $e_6$ |

$E^+$

| | k-1 | k | k+1 |
|---|---|---|---|
| j-1 | $e_6$ | $e_5$ | $e_4$ |
| j | $-e_2$ | $e_3$ | $e_2$ |
| j+1 | 0 | $e_1$ | 0 |

where

$$e_1 = \frac{1}{2}\left(\frac{1}{\Delta z \Delta t} + \frac{b_{11}}{\Delta t^2}\right), \qquad e_2 = \frac{1}{2}\left(\frac{a_{11}}{\Delta z \Delta x} + \frac{c_{11}}{2\Delta t \Delta x}\right)$$

$$e_3 = -\frac{b_{11}}{\Delta t^2}, \qquad e_4 = \frac{1}{2}\left(\frac{d_{11}}{\Delta x^2} - \frac{c_{11}}{2\Delta t \Delta x}\right)$$

$$e_5 = \frac{1}{2}\left(\frac{b_{11}}{\Delta t^2} - \frac{1}{\Delta z \Delta t}\right) - \frac{d_{11}}{\Delta x^2}$$

$$e_6 = \frac{1}{2}\left(\frac{c_{11}}{2\Delta t \Delta x} + \frac{d_{11}}{\Delta x^2}\right)$$

( $a_{11}$, $b_{11}$, $c_{11}$ and $d_{11}$ denote the (1,1) element of the matrices A, B, C and D respectively)

The F, G and H matrices are found by using subscripts (1,2), (2,1) and (2,2) respectively for the a, b, c and d coefficients in (4). The only exception is that the non-coefficient terms (i.e., $1/\Delta z\Delta t$ ) are omitted in F and H .

At each point in the two components of equation (2) the ($e_1$, $f_1$, $g_1$, and $h_1$ ) terms of the '+' matrices are the coefficients of the prediction points. Hence, we have a 2x2 system of the following form

$$\begin{aligned} e_1 u' + f_1 w' &= r_1 \\ g_1 w' + h_1 u' &= r_2 \end{aligned} \tag{4}$$

to solve for the new values $u'$ ( $= u^{n+1}_{j,k}$ ) and $w'$ ( $= w^{n}_{j,k}$ ) . The residue terms $r_1$ and $r_2$ are composed of all other sums of the E and F patterns and the G and H patterns respectively.

We will solve for the point ( j+1, k, n+1 ) in each step. That point is not coupled by any x differencing and hence the scheme is explicit. It is also straightforward to see that the error in formula (2) is of the order $0 ( \Delta t^2 + \Delta x^2 + \Delta z^2 )$ . This is due to the centering and is checked by inserting the true solution to (1) into the difference equation and expanding in a Taylor series (compare SEP-7, p. 147, 148).

A thorough stability analysis of (2) is very complicated. Even checking stability of the pure initial value problem in z and the stability of one sweep in t for a fixed z level involves studying the roots of generalized eigenvalue problems. Hence, we will consider only the simpler case of stability for the scalar analog of (2).

This will give us necessary conditions for bounded solutions in the scalar case and also insight regarding the stability questions of the original problem. (The index $k$ below is omitted.)

$$u_{tz} + a\, u_{xz} + b\, u_{tt} + c\, u_{tx} + d\, u_{xx} = 0 \qquad (6)$$

$$D_0^t D_+^z u_j^n + a\, D_0^x D_+^z u_j^n + \frac{b}{2} D_+^t D_-^t ( u_j^n + u_j^{n+1} ) +$$
$$+ \frac{c}{2} D_+^t D_0^x ( u_j^n + u_{j-1}^{n+1} ) + \frac{d}{2} D_+^x D_-^x ( u_{j+1}^n + u_{j-1}^{n+1} ) = 0 \qquad (7)$$

The stability in the $z$ direction when neglecting the boundary in time $(t \geq 0)$ follows easily. After transforming in $x$ and $t$ we get from (7)

$$\hat{u}^{n+1} = \beta ( \Delta t \omega, \Delta x\, k_x ) \hat{u}^n$$

where $|\beta| = 1$ . [The calculations are straightforward (compare SEP-7, p. 141-143) and are omitted.] Hence, the scheme is energy conserving. The analogous result for (6) is

$$\hat{u}_z = \alpha ( \omega, k_x ) \hat{u}$$

where the real part of $\alpha$ vanishes.

The next check is on the stability of (7) for one sweep in $t$ when $z$ is fixed. To do this we drop all terms with index $n$ and transform in $x$ . (The $n+1$ index is omitted to simplify notation.)

$$( u_{j+1} - u_{j-1} ) + i \bar{a} \sin\xi \, u_j + \bar{b} ( u_{j+1} - 2 u_j + u_{j-1} ) +$$

$$+ i \bar{c} \sin \xi ( u_j - u_{j-1} ) - \bar{d} \sin^2 \xi/2 \; u_{j-1} = 0 \qquad (8)$$

$$\xi = \Delta x \, k_x \quad , \quad \bar{a} = 2 \frac{\Delta t}{\Delta x} a \quad , \quad \bar{b} = \frac{\Delta z}{\Delta t} b$$

$$\bar{c} = \frac{\Delta z}{\Delta x} c \quad , \quad d = 4 \frac{\Delta t \Delta z}{\Delta x^2} d$$

The difference equation (8) is stable if the corresponding characteristic polynomial has no roots outside the unit circle. The polynomial is

$$( s^2 - 1 ) + i \bar{a} \sin\xi \, s + \bar{b} ( s^2 - 2 s + 1 ) +$$

$$i \bar{c} \sin \xi ( s-1) - \bar{d} \sin^2 \xi/2 = 0 \qquad (9)$$

With $A = \bar{a} \sin \xi$ , $B = \bar{b}$ , $C = \bar{c} \sin \xi$ , $D = \bar{d} \sin^2 \xi/2$ , we have

$$s^2 + \frac{i(A+C) - 2B}{1+B} s + \frac{-1 + B - i C - D}{1+B} = 0 \qquad (10)$$

For the differential equation (6) to be well posed with time as the evolution direction the range of the coefficients has to be restricted. We will make the following assumptions for the scalar equation which are similar to the ones we have in the vector case used in the computations.

$$b \geq 0 \quad , \quad d \leq 0$$

$$a c \geq \varepsilon a^2 \quad , \quad \varepsilon > 0 \qquad (11)$$

$$\text{if} \quad b=0 \quad \text{then} \quad a=0, \; d < 0$$

With these conditions we can prove the <u>proposition</u>: For a fixed $\Delta z/\Delta t$ which is small enough we can choose $\Delta x/\Delta z$ large enough so that (7) is stable for one sweep in $t$ when $z$ is fixed. This means that the stability condition looks like

$$\frac{\Delta z}{\Delta t} \leq C_1 \quad , \quad \frac{\Delta t}{\Delta x} \leq C_2$$

The actual values of the constants $C_1$ and $C_2$ for the full equation (2) are best determined through practical tests with initial data containing high frequencies.

<u>Proof of Proposition</u>: We want to show that the roots of (10) are not outside the unit circle

$$s = -\frac{i(A+C)}{2(1+B)} + \frac{B}{1+B} \pm \left( -\frac{(A+C)^2}{4(1+B)^2} - \frac{i(A+C)B}{(1+B)^2} + \frac{B^2}{(1+B)^2} + \frac{1-B+iC+D}{1+B} \right)^{1/2} \tag{12}$$

1) $B = 0$: From the assumption we have $A = 0$.

$$s = -\frac{iC}{2} (\pm) \left( 1 - \frac{C^2}{4} + iC + D \right)^{1/2} \tag{13}$$

We Taylor expand the square and use the minus sign, which corresponds to the largest root.

$$s = -\frac{iC}{2} - \left( 1 - \frac{C^2}{8} + \frac{iC}{2} + \frac{D}{2} + \frac{C^2}{8} + \cdots \right)$$

$$|s|^2 = (\text{real part})^2 + (\text{imaginary part})^2 \leq$$

$$\leq (1+D) + (C^2) + \text{higher order terms}$$

With higher order terms we mean terms that decay faster when $\frac{\Delta z}{\Delta t}$ and $\frac{\Delta t}{\Delta x}$ become smaller. Here $|s| \leq 1$ when $\frac{\Delta z}{\Delta t}$ is small enough, since then the negative D will dominate $C^2$ . Assumption (11) implies $d < 0$ .

$$D = \frac{4\,\Delta t\,\Delta z}{\Delta x^2} \sin^2 \xi/2\, d \ , \quad C^2 = \frac{\Delta z^2}{\Delta x^2} \sin^2 \xi\, c^2$$

2) $B > 0$ : If A , C and D are small enough the root corresponding to the plus sign is the critical one.

$$s = -\frac{i(A+C)}{2(1+B)} + \frac{B}{1+B} + \left( \left( \frac{1+\frac{iC}{2}}{1+B} \right)^2 - \frac{A^2+2AC}{4(1+B)^2} - \frac{i\,AB}{(1+B)^2} + \frac{D}{1+B} \right)^{1/2}$$

$$s = -\frac{i(A+C)}{2(1+B)} + \frac{B}{1+B} + \left( \frac{1+\frac{iC}{2}}{1+B} \right) \left( 1 + \frac{1}{4(1+\frac{iC}{2})^2} \left( -(A^2+2\,AC) - 4i\,AB + 4\,D(1+B) \right) \right)^{1/2} \qquad (14)$$

We Taylor expand the square root again about 1 .

$$s = 1 - \frac{i\,A}{2(1+B)} + \left( \frac{1+\frac{iC}{2}}{1+B} \right) \left( \frac{1}{8(1+\frac{iC}{2})^2} \left( -(A^2+2AC) - 4\,i\,A\,B + 4D(1+B) \right) + \dots \right)$$

$|s|^2 = (\text{real part})^2 + (\text{imaginary part})^2 \leq 1 - \frac{AC}{2}$ + terms of higher order than $A^2$

The condition $a\,c \geq \varepsilon\, a^2$ implies that $|s| \leq 1$ , and our proof is concluded.

This type of analysis gives an understanding of the stability properties and is of help in the design of the difference scheme. We can, for example, see that it is risky to approximate the $U_{xx}$ term by the simple centered approximation $D_+^x D_-^x ( U_{j,k}^n + U_{j,k}^{n+1} )$. If we consider the Nyquist frequency ($\xi = \pi$) the factor $\sin\xi$ vanishes and the equation corresponding to (9) looks like

$$( s^2 - 1 ) + \bar{b} ( s^2 - 2s + 1 ) - \bar{d} s = 0 \tag{15}$$

The negative root becomes larger than one in magnitude when $\bar{b}$ approaches zero.

$$s^2 - \bar{d} s - 1 = 0$$

$$s = \frac{\bar{d}}{2} (\pm) \left( 1 + \left(\frac{\bar{d}}{2}\right)^2 \right)^{1/2} \leq - \left( 1 + \left|\frac{\bar{d}}{2}\right| \right)$$

When the B and D matrices in (1) and (2) are diagonal, which is the case in our experiments, this argument is rigorous. For the use of retarded time ($t' = t - b_{22} z$) the scheme must work for $b = 0$.

Let us comment on some variants of our approach for future applications, before the discussion of the numerical experiments.

There is the possibility of using different matrices A, B, C and D corresponding to different orders of approximation, the use of retarded time or corresponding to different sets of dependent variables. One can, for example, eliminate the terms containing mixed derivatives. We will however, concentrate on two numerical techniques: 1) the use of implicit schemes and 2) splitting.

1) By other types of averaging in the difference approximation we can derive implicit schemes where at least in certain cases we get rid of the stability limits on $\Delta t$, $\Delta x$ and $\Delta z$. Implicitness means here that $x$ differencing occurs also in the last $t$ and $z$ levels. This coupling in the $x$ direction forces us to solve a block tridiagonal system of linear equations for each new $t$ step. The blocks are just 2x2 so that the solution to such a system should not be too costly. The most straightforward scheme is the following Crank-Nicolson-like formula. (The $k$ index is omitted.)

$$D_0^t D_+^z U_j^n + \frac{A}{2} D_0^x D_+^z ( U_{j+1}^n + U_{j-1}^n ) + \frac{B}{2} D_+^t D_-^t ( U_j^n + U_j^{n+1} ) + \\ + \frac{C}{2} D_0^t D_0^x ( U_j^n + U_j^{n+1} ) + \frac{D}{4} D_+^x D_-^x ( U_{j-1}^n + U_{j+1}^n + U_{j-1}^{n+1} + U_{j+1}^{n+1} ) = 0 \qquad (16)$$

Within the same difference molecule we can change the $x$ differences $D_+^x D_-^x$ and $D_0^x$ to $D_+^x D_-^x ( I + \alpha \Delta x^2 D_+^x D_-^x )^{-1}$ and $D_0^x ( I + \alpha \Delta x^2 D_+^x D_-^x )^{-1}$ and get better accuracy (reduce numerical dispersion) with a proper choice of the scalar $\alpha$. This is particularly efficient when $A$ and $C$ vanishes in the formula.

2) Splitting has been successfully applied by Claerbout to the scalar problem (see Claerbout, p. 254).

$$P_{tz} = - \frac{1}{v} P_{tt} + \frac{v}{2} P_{xx} \qquad (17)$$

The idea is that each $z$ step is divided into two parts. First we solve $P_{tz} = - \frac{1}{v} P_{tt}$ ( or $P_z = - \frac{1}{v} P_t$ ) for one $\Delta z$ step starting from the known values $P^n$. Then, using the new $P$ values, the equation

$P_{tz} = \frac{v}{2} P_{xx}$ is solved for a $\Delta z$ step to get $P^{n+1}$ . The advantage is here that each smaller equation is much easier to solve than (17). The problem $P_z = -\frac{1}{v} P_t$ will just amount to a shift. $P(t,x,z+\Delta z) = P(t + \frac{\Delta z}{v}, x, z)$ .

We will have the same advantage for the elastic wave equation if, for example, $U_{tz} + BU_{tt} = 0$ is solved separately by a shift. It is a convenient way to handle the two different wave speeds. Since the matrices A , B , C and D in general do not commute there will be an error of order $O(\Delta z)$ if the splitting is performed in the same fashion as the scalar case. With some extra care this can be avoided. In the equation (1) we can, for example, first make a shift (corresponding to $U_{tz} + BU_{xx} = 0$ ) of $\Delta z/2$ then solve $U_{tz} + AU_{xz} + CU_{tx} + DU_{xx} = 0$ for a step $\Delta z$ and finally a shift of $\Delta z/2$ to set the values on the new z level. In the interior of the z interval the two shifts of half a stepsize can be combined to make one full sized step.

Let us in order to simplify the arguments apply splitting to an ordinary differential equation with matrix coefficients, and look at the effect of non-commutativity.

$$U_z = EU + FU \tag{18}$$

Its solution for a $\Delta z$ interval is

$$U(z + \Delta z) = e^{(E+F)\Delta z} U(z) =$$

$$= ( I + \Delta z(E+F) ) + \frac{\Delta z^2}{2} ( E^2 + F^2 + EF + FE) + O(\Delta z^3) ) U(z) \tag{19}$$

The scalar technique of first solving $U_z = EU$ and then $U_z = FU$ gives

$$U_1(z+\Delta z) = e^{F\Delta z} e^{E\Delta z} U(z) =$$

$$= \left( I + \Delta z(E+F) + \frac{\Delta z^2}{2}(E^2 + F^2 + 2FE) + O(\Delta z^3) \right) U(z)$$

When $E$ and $F$ do not commute we have an $O(\Delta z^2)$ error in each step. After $z_{max}/\Delta z$ steps this can add up to an $O(\Delta z)$ error. If we instead use the proposed splitting method the approximation agrees to order $O(\Delta z^3)$ with (19) .

$$U_2(z+\Delta z) = e^{E\frac{\Delta z}{2}} e^{F\Delta z} e^{E\frac{\Delta z}{2}} U(z) =$$

$$= \left( I + \Delta z(E+F) + \frac{\Delta z^2}{2}(E^2 + F^2 + EF + FE) + O(\Delta z^3) \right) U(z)$$

<u>Numerical Example</u>

The derivation of the previous section was tested with a simple numerical example. We chose for the test, the following coefficient matrices for equation (1) [see p. 129 in this report].

$$A = \mathbf{0} \qquad B = \begin{pmatrix} \frac{1}{\sqrt{d_2}} & 0 \\ 0 & \frac{1}{\sqrt{d_1}} \end{pmatrix}$$

$$C = \gamma \begin{pmatrix} 0 & \frac{1}{\sqrt{d_2}} \\ \frac{1}{\sqrt{d_1}} & 0 \end{pmatrix} \qquad D = \frac{1}{2} \begin{pmatrix} (\gamma^2-d_1)/\sqrt{d_2} & 0 \\ 0 & (\gamma^2-d_2)/\sqrt{d_1} \end{pmatrix}$$

where $d_1 = \lambda + 2\mu$ , $d_2 = \mu$ , $\gamma = \frac{\lambda+\mu}{\sqrt{d_1}+\sqrt{d_2}}$

The following elastic and grid constants were used:

$$\lambda = 1.0 \qquad \mu = 1.0 \qquad \rho = 1.0$$

$$\Delta x = 1.0 \qquad \Delta t = 0.3 \qquad \Delta z = 0.1$$

The initial conditions were specified as various combinations of smoothed impulses in the $u$ and $w$ fields. In Figures 2 and 3 the impulse is in the $u$ field with a zero $w$ field. Figures 4 and 5 are the opposite case, and Figures 6 and 7 have impulses in both fields. Figure 1 shows the impulse and also serves as a prototype of the plot format in the following figures.




Figure 1. The impulse function.

The migration was allowed to proceed through 40 steps of extrapolation in the $z$ direction, and plots are presented for the $20^{th}$ and $40^{th}$ steps. Each figure shows both the $u$ and $w$ fields, each with two types of scaling. The absolute scaling (same scale constant used throughout) shows how the energy changes as a function of $z$. Relative scaling (each plot individually scaled) gives a better idea of the diffraction shape for each frame.

Figures 2 through 5 are relatively easy to interpret. In each case, the coupling terms give rise to an asymmetrical hyperbolic diffraction pattern in the field that was initially specified as zero. In other words, a displacement in one field gives rise to a displacement in the other field which travels in opposite directions about the line of symmetry. Also in each case, the $u$ field moves further in time and has a broader hyperbola than the $w$ field, as one would expect.

In the last set of figures (6 and 7), the initial impulse in each field gives rise to diffraction patterns which are a linear combination of the single impulse patterns.

The value $\Delta x = 0.5$ still leads to a stable approximation for $\Delta t = 0.3$ and $\Delta z = 0.1$, but $\Delta x = 0.25$ produces a strong instability. Tests also showed that the transformation to retarded time ( $t' = t - z / \sqrt{d_1}$ ) did work as expected.

A listing of the elastic equation subroutine used in this example follows the figures. In writing this program a "flexible" approach rather than an "efficient" approach was taken.




Figure 2. Impulse in u , n = 20 .




Figure 3. Impulse in u , n = 40 .




Figure 4. Impulse in w, $n = 20$.




Figure 5. Impulse in w , n = 40 .




Figure 6. Impulses in u and w , n = 20 .




Figure 7. Impulses in u and w , n = 40 .

```
C   EWAVE
C
C   A SUB/R FOR A SINGLE STEP MIGRATION OF ELASTIC WAVES
C   DESCRIBED INTERMS OF U -HORIZONTAL DISP. AND
C   W -VERTICAL DISP.
C
C   (BU,BW) ARE THE U & W FIELDS AT THE PREVIUOS Z STEP
C   (AU,AW) ARE THE U & W FIELDS OUTPUT AT THE NEW Z STEP.
C   (BU,BW) AND (AU,AW) MUST BE SWAPPED WITH EACH OTHER
C    BY THE CALLING PROGRAM.
C
C   (SU & SW) ARE VECTOR BUFFERS ATTACHED TO THE OLD X-T FRAME.
C   (TU & TW) ARE VECTOR BUFFERS ATTACHED TO THE NEW X-T FRAME
C
C   THE POINTERS IP1...3 POINT TO THE FOLLOWING
C            IP1 -> CURRENT X-STRIP
C            IP2 -> 1 TIME STEP OLD X-STRIP
C            IP3 -> 2 TIME STEPS OLD X-STRIP
C
        SUBROUTINE EWAVE(AU,AW,BU,BW,NDIM,NT,NX)
        REAL AU(NDIM,NT),AW(NDIM,NT),BU(NDIM,NT),BW(NDIM,NT)
        REAL TU(50,3),TW(50,3),SU(50,3),SW(50,3)
        COMMON/EFGH/ E1,E2,E3,E4,E5,E6,
       1             F1,F2,F3,F4,F5,F6,
       2             G1,G2,G3,G4,G5,G6,
       3             H1,H2,H3,H4,H5,H6
        NX1=NX-1
        DET= H1*F1 -E1*G1
C   SET INITIAL POINTERS
        IP1=1
        IP2=2
        IP3=3
C   SET INITIAL STRIPS TO ZERO
        DO 10 I=1,NX
        SU(I,2)=0.
        SU(I,3)=0.
        TU(I,2)=0.
        TU(I,3)=0.
        TW(I,2)=0.
        TW(I,3)=0.
        SW(I,2)=0.
    10  SW(I,3)=0.
C   MAIN LOOP OVER X-STRIPS
        DO 50 J=1,NT
C   LOOP OVER EACH X-STRIP
        DO 20 I=1,NX
        SU(I,IP1)= BU(I,J)
    20  SW(I,IP1)= BW(I,J)
C
```

```
      DO 30 I=2,NX1
C
C   APPLY PATTERN MATRICES E, F, G AND H.
C
      R1=  E1*(SU(I,IP3))
     1     +E2*(SU(I-1,IP2)-SU(I+1,IP2)-TU(I-1,IP2)+TU(I+1,IP2))
     2     +E3*(SU(I,IP2)+TU(I,IP2))
     3     +E4*(SU(I-1,IP1)+TU(I+1,IP3))
     4     +E5*(SU(I,IP1)+TU(I,IP3))
     5     +E6*(SU(I+1,IP1)+TU(I-1,IP3))
     6     +F1*(SW(I,IP3))
     7     +F2*(SW(I-1,IP2)-SW(I+1,IP2)-TW(I-1,IP2)+TW(I+1,IP2))
     8     +F3*(SW(I,IP2)+TW(I,IP2))
     9     +F4*(SW(I-1,IP1)+TW(I+1,IP3))
     A     +F5*(SW(I,IP1)+TW(I,IP3))
     B     +F6*(SW(I+1,IP1)+TW(I-1,IP3))
      R2=  G1*(SW(I,IP3))
     1     +G2*(SW(I-1,IP2)-SW(I+1,IP2)-TW(I-1,IP2)+TW(I+1,IP2))
     2     +G3*(SW(I,IP2)+TW(I,IP2))
     3     +G4*(SW(I-1,IP1)+TW(I+1,IP3))
     4     +G5*(SW(I,IP1)+TW(I,IP3))
     5     +G6*(SW(I+1,IP1)+TW(I-1,IP3))
     6     +H1*(SU(I,IP3))
     7     +H2*(SU(I-1,IP2)-SU(I+1,IP2)-TU(I-1,IP2)+TU(I+1,IP2))
     8     +H3*(SU(I,IP2)+TU(I,IP2))
     9     +H4*(SU(I-1,IP1)+TU(I+1,IP3))
     A     +H5*(SU(I,IP1)+TU(I,IP3))
     B     +H6*(SU(I+1,IP1)+TU(I-1,IP3))
C
C   SOLVE 2X2 SYSTEM FOR NEW U AND W POINTS
C
      TU(I,IP1)= (G1*R1 - F1*R2)/DET
      TW(I,IP1)= (E1*R2 - H1*R1)/DET
C
   30 CONTINUE
C   SET BOUNDARY CONDITIONS
      TU(1,IP1)= TU(2,IP1)
      TU(NX,IP1)= TU(NX1,IP1)
      TW(1,IP1)= TW(2,IP1)
      TW(NX,IP1)= TW(NX1,IP1)
C
      DO 40 I=1,NX
      AU(I,J)= TU(I,IP1)
   40 AW(I,J)= TW(I,IP1)
C   ROTATE POINTERS
      ITEMP= IP3
      IP3= IP2
      IP2= IP1
      IP1= ITEMP
C
   50 CONTINUE
C
      RETURN
      END
```

References:

[1] Kelly, K. R., R. W. Ward, Sven Treitel and R. M. Alford: Synthetic seismograms: A finite-difference approach, Geophysics, vol. 41, no. 1, pp. 2-27, 1976.

[2] Claerbout, Jon F.: Fundamentals of Geophysical Data Processing: With Application to Petroleum Prospecting, McGraw-Hill, Inc., 1976.